
;***********************************
;Programa: lcd1.asm
;Autor: Rogério M. Branco
;Compilador/Montador: Keil 6.20
;Data: 25/11
;***********************************
;Função: Enviar uma mensagem p/ LCD
;***********************************
;Considerações: 
;       
;        LCD de 14 pinos/ 
;        comunicação 8 bits/ 
;        controlador Hitachi hd44780 
;        ou smos sed1130
;
;***********************************
;Pinagem do LCD:
;
;Pino   Nome    Ligação Função
; 01    Vss     GND     Terra
; 02    Vcc     +5v     Alimentação
; 03    Vee     AJT     Controle de conraste
; 04    RS      1/0     Dado=1; Instrução=0;
; 05    R/(/W)  1/0     1=Read=Lcd->CPU; 0=Write=CPU->Lcd
; 06    E       CS      Habilita Oper. LCD (qdo 1 ou 0->1)
; 07    DB0     DADO    dado menos significativo
; 08    DB1     DADO    dado
; 09    DB2     DADO    dado
; 10    DB3     DADO    dado
; 11    DB4     DADO    dado
; 12    DB5     DADO    dado
; 13    DB6     DADO    dado
; 14    DB7     DADO    dado mais significativo
;***********************************
;Resumo de Instruções do display:
;       Para usar, enviar:      RS=0 (instr)
;                               Envia palavra
;                               E=1, E=0
;
DISP_ON_S_CURS  EQU      0CH
DISP_ON_C_CURS  EQU      0EH
DISP_OFF        EQU      08H

CLRLCD          EQU      01H

CURSOR_C1L1     EQU      80H    
CURSOR_C1L2     EQU     0C0H

ESCR_CURS_DIR   EQU      06H
ESCR_CURS_ESQ   EQU      04H

CONF_8b_1L_5X7  EQU      30H
CONF_8b_2L_5X7  EQU      38H
CONF_8b_1L_5X10 EQU      34H

CONF_4b_1L_5X7  EQU      20H
CONF_4b_2L_5X7  EQU      28H
CONF_4b_1L_5X10 EQU      24H
;
;
;***********************************
;DEFINIÇÃO DE PINOS 
;
LCD_E           EQU     P3.7
LCD_RS          EQU     P3.6
LCD_R_W         EQU     P3.0
;
;

;***********************************
;PROGRAMA ASM
;***********************************
;END. DAS INTERRUPÇÕES VETORADAS
;***********************************
        ORG 0000H
        LJMP INICIO
;***************************
;INICIO DO ROGRAMA PRINCIPAL
;***************************
        ORG 0100H
INICIO: LCALL INI_UC
        LCALL INI_LCD
        LCALL POS_L1
        MOV DPTR,#FRASE1
        LCALL ESCREVE
        LCALL POS_L2
        MOV DPTR,#FRASE2
        LCALL ESCREVE
        JMP $
;***************************
;AREA DAS FUNÇÕES UTILIZADAS
;***************************
WR_INST:CLR LCD_RS      ;RS=0 PARA INSTRUÇÃO
        CLR LCD_R_W     ;R/W\=0 PARA ESCRITA
        MOV P1,A        ;DADO DO ACC PARA P1=INSTRUÇÃO DISPLAY
        SETB LCD_E      ;ATIVA E DO LCD
        CLR LCD_E       ;DESATIVA E DO LCD      
        LCALL OCUPADO   ;E AGUARDA ATÉ DESOCUPAR        
        RET             

RD_INST:CLR LCD_RS      ;RS=0 PARA INSTRUÇÃO
        SETB LCD_R_W    ;R/W\ =1 PARA LEITURA
        SETB LCD_E      ;ATIVA E DO LCD
        CLR LCD_E       ;DESATIVA E DO LCD      
        MOV A,P1        ;DADO DE P1 PARA ACC
        RET

WR_DADO:SETB LCD_RS     ;RS=1 PARA DADO
        CLR LCD_R_W     ;R/W\ = 0 PARA ESCRITA
        MOV P1,A        ;DADO DO ACC PARA P1=DADOS DISPLAY
        SETB LCD_E      ;ATIVA E DO LCD
        CLR LCD_E       ;DESATIVA E DO LCD      
        LCALL OCUPADO   ;E AGUARDA ATÉ DESOCUPAR        
        RET             

RD_DADO:SETB LCD_RS     ;RS=1 PARA DADO
        SETB LCD_R_W    ;R/W\ =1 PARA LEITURA
        MOV P1,A        ;DADO DO ACC PARA P1=DADOS DISPLAY
        SETB LCD_E      ;ATIVA E DO LCD
        CLR LCD_E       ;DESATIVA E DO LCD      
        RET             

OCUPADO:ACALL RD_INST
        JB ACC.7,OCUPADO
        RET

POS_L1: MOV A,#CURSOR_C1L1      ;CARREGA INSTR POSIC CURSOR C1, L1
        LCALL WR_INST
        RET

POS_L2: MOV A,#CURSOR_C1L2      ;CARREGA INSTR POSIC CURSOR C1, L2
        LCALL WR_INST
        RET

INI_UC: MOV P0,#0
        MOV P1,#0
        MOV P2,#0

        MOV P3,#0
        CLR EA          ;DESABILITA INTERRUPÇÕES
        RET

CLR_LCD:MOV A,#CLRLCD   ;ESCREVE INST DE LIMPEZA DO LCD
        LCALL WR_INST   ;
        RET

INI_LCD:MOV A,#CONF_8b_2L_5X7   ;LCD 16*2, 8 BITS, 5*7  
        LCALL WR_INST           ;

        MOV A,#CONF_8b_2L_5X7   ;LCD 16*2, 8 BITS, 5*7  
        LCALL WR_INST           ;
        
        MOV A,#DISP_ON_C_CURS   ;LCD LIGADO COM CURSOR ATIVO
        LCALL WR_INST           ;

        MOV A,#ESCR_CURS_DIR    ;LCD DESLOC CURSOR P/ DIREITA  
        LCALL WR_INST           ;

        LCALL CLR_LCD
        RET

ESCREVE:CLR A
        MOVC A,@A+DPTR
        JZ SAIR
        LCALL WR_DADO
        INC DPTR
        AJMP ESCREVE
SAIR:   NOP
        RET

;***************************
;AREA DE DADOS DAS FRASES
;***************************
FRASE1: DB "* CURSO DE POS *"
        DB 00H

FRASE2: DB "* GRADUAÇÃO AI *"
        DB 00H
        
;***************************
;FIM
;***************************
        END

